

```

Data repeated;
  infile
"C:\Users\edwardsm\Documents\Workshops\SAS\Level_III\Repeated\Repeated.csv"
dlm = "," firstobs =2 missover;
  input ID    room    trmt        day        wt;
Run;

Data repeated_mult;
  infile
"C:\Users\edwardsm\Documents\Workshops\SAS\Level_III\Repeated\Repeated_mult.c
sv" dlm = "," firstobs =2 missover;
  input ID          trmt        day        wt1 wt2 wt3;
Run;

*ods pdf
file="C:\Users\edwardsm\Documents\Workshops\SAS\Level_III\RepeatedMeasures_20
170531.pdf";

Proc print data=repeated;
  title "Univariate form of a Repeated Measures trial";
Run;

Proc print data=repeated_mult;
  title "Multivariate form of a Repeated Measures trial";
Run;

/* Repeated Measures as a Split plot design - Trmt as main plot and day as
subplot - Proc GLM */

Proc glm data=repeated;
  class trmt day ID;
  model wt = trmt ID(trmt)  day day*trmt;
  test h=trmt e=ID(trmt);
  title "Repeated Measures as a Split plot design - Trmt as main plot and day
as subplot - Proc GLM";
Run;
Quit;

/* Repeated Measures using the REPEATED statement and multivariate format of
the dataset - Proc GLM */

Proc glm data=repeated_mult;
  class trmt ID;
  model wt1-wt3 = trmt / nouni;
  repeated time polynomial / printe summary;
  title "Repeated Measures using the REPEATED statement and multivariate
format of the dataset - Proc GLM";
Run;
Quit;

/* Repeated Measures using Proc MIXED */

Proc mixed data=repeated covtest;
  class trmt day;
  model wt = trmt|day;

```

```

    repeated day / subject = ID type=cs;
    title "Repeated Measures using Proc MIXED";
Run;

/* Repeated Measures using Proc GLIMMIX */

Proc glimmix data=repeated;
    class trmt day;
    model wt = trmt|day;
    random _residual_ / subject = ID type=cs;
    title "Repeated Measures using Proc GLIMMIX";
Run;

/* Repeated Measures using Proc GLIMMIX with a random effect (room) */

Proc glimmix data=repeated;
    class trmt day;
    model wt = trmt|day;
    random room;
    random _residual_ / subject = ID type=cs;
    title "Repeated Measures using Proc GLIMMIX with RANDOM variable (room)";
Run;

Proc glimmix data=repeated;
    class trmt day;
    model wt = trmt|day;
    random room;
*   random _residual_ / subject = ID type=cs;
    title "Removing the Repeated statement - Proc GLIMMIX with RANDOM variable (room)";
Run;

Proc glimmix data=repeated;
    class trmt day;
    model wt = trmt|day;
*   random room;
    random _residual_ / subject = ID type=cs;
    title "Removing the Random Room statement - Proc GLIMMIX with RANDOM variable (room)";
Run;

*ods pdf close;

Proc glimmix data=repeated;
    class trmt day;
    model wt = trmt|day;
*   random room;
    random _residual_ / subject = ID type=cs;
    output out=second predicted=pred residual=resid residual(noblup)=mresid
student=studentresid student(noblup)=smresid;
    title "Removing the Random Room statement - Proc GLIMMIX with RANDOM variable (room)";
Run;

title;

```

```
/* Linearity of fixed effects - both a scatter and a boxplot */
Proc sgplot data=second;
    scatter y=smresid x = day;
    refline 0;
Run;

Proc sgplot data=second;
    vbox smresid / group=day datalabel;
Run;

/* Homogeneity of effects */
Proc sgscatter data=second;
    plot studentresid*(pred day);
Run;

/* Q-Q plot of normal distribution */
Proc univariate data=second normal;
    var resid mresid;
Run;
```